PC8563 Arduino interface i am trying to interface pc8563 with arduino uno . I have small doubt related to RTC slave address reading

RTC datasheet says page no 14 
where device slave address is A2 for write and Read A3 
i would like to know how A2 /A3

Example code from website device id mentioned as 0X51  

My question how to config exact slave address ??

#include "Wire.h"
#define PCF8563address 0x51

byte second, minute, hour, dayOfWeek, dayOfMonth, month, year;
String days[] = {"Sunday", "Monday", "Tuesday", "Wednesday", "Thursday", "Friday", "Saturday" };

byte bcdToDec(byte value)
{
  return ((value / 16) * 10 + value % 16);
}

byte decToBcd(byte value){
  return (value / 10 * 16 + value % 10);
}

void setPCF8563()
// this sets the time and date to the PCF8563
{
  Wire.beginTransmission(PCF8563address);
  Wire.write(0x02);
  Wire.write(decToBcd(second)); 
  Wire.write(decToBcd(minute));
  Wire.write(decToBcd(hour));     
  Wire.write(decToBcd(dayOfMonth));
  Wire.write(decToBcd(dayOfWeek)); 
  Wire.write(decToBcd(month));
  Wire.write(decToBcd(year));
  Wire.endTransmission();
}

void readPCF8563()
// this gets the time and date from the PCF8563
{
  Wire.beginTransmission(PCF8563address);
  Wire.write(0x02);
  Wire.endTransmission();
  Wire.requestFrom(PCF8563address, 7);
  second     = bcdToDec(Wire.read() &amp; B01111111); // remove VL error bit
  minute     = bcdToDec(Wire.read() &amp; B01111111); // remove unwanted bits from MSB
  hour       = bcdToDec(Wire.read() &amp; B00111111);
  dayOfMonth = bcdToDec(Wire.read() &amp; B00111111);
  dayOfWeek  = bcdToDec(Wire.read() &amp; B00000111); 
  month      = bcdToDec(Wire.read() &amp; B00011111);  // remove century bit, 1999 is over
  year       = bcdToDec(Wire.read());
}

void setup()
{
  Wire.begin();
  Serial.begin(9600);
  // change the following to set your initial time
  second = 0;
  minute = 28;
  hour = 9;
  dayOfWeek = 2;
  dayOfMonth = 13;
  month = 8;
  year = 13;
  // comment out the next line and upload again to set and keep the time from resetting every reset
  setPCF8563();
}

void loop()
{
  readPCF8563();
  Serial.print(days[dayOfWeek]);
  Serial.print(" "); 
  Serial.print(dayOfMonth, DEC);
  Serial.print("/");
  Serial.print(month, DEC);
  Serial.print("/20");
  Serial.print(year, DEC);
  Serial.print(" - ");
  Serial.print(hour, DEC);
  Serial.print(":");
  if (minute &lt; 10)
  {
    Serial.print("0");
  }
  Serial.print(minute, DEC);
  Serial.print(":"); 
  if (second &lt; 10)
  {
    Serial.print("0");
  } 
  Serial.println(second, DEC); 
  delay(1000);
}


RTC datasheet
Normal I2C addresses are 7-bit long. On the first frame of an I2C
transaction, the master sends an 8-bit byte containing the address of
the slave followed by a bit specifying whether this is a read or a write
operation. Some people tend to call “address” the whole 8-bit frame,
which causes confusion. Because of this, we end up having three flavors
of I2C addresses:


the 7-bit address, which is the real device address
the 8-bit “write address”, which is the address followed by a 0
meaning “write operation”
the 8-bit “read address”, which is the address followed by a 1
meaning “read operation”


For your RTC, the 3 addresses are the following, in hex and binary:

7-bit = 0x51 = 0b1010001
write = 0xA2 = 0b10100010
read  = 0xA3 = 0b10100011


The Arduino Wire library expects the 7-bit address.